Using Maxima to Calculate the Stable Phases of Solid Plutonium

Pure plutonium fors six different equilibrium solids at atmospheric pressure. Solid Pu can have six different crystal structures depending on temperature, therefore there are six solid phases given the following names: α, β, γ δ, δ' and ϵ. In this section, we will use Maxima to generate plots of the molar enthalpy, entropy and Gibbs free energy as functions of temperature for the stable phases of solid plutonium. The resulting Gibbs diagram will clearly indicate the stable phase for a given temperature. We will ignore the δ' phase, since it is stable over a very narrow temperature range at atmospheric pressure.

The enthalpy, entropy and Gibbs free energy can be calculated from the following:

$$\overline{H}(T) = \int_{T_i}^{T_f} \overline{C}_p dT \tag{1}$$

$$\overline{S}(T) = \int_{T_i}^{T_f} \frac{\overline{C}_p}{T} dT \tag{2}$$

$$\overline{G}(T) = \overline{H}(T) - T\,\overline{S}(T) \tag{3}$$

The heat capacity at constant pressure for Pu is a weak function of temperature for the α, β and γ phases and is constant for the δ and ϵ phases. Heat capacities in units of $\frac{cal}{K\,mol}$ are given as follows for the various temperature ranges:

The heat capacities were taken from the scientific literature on Pu.

$$\overline{C}_p(T) = \begin{cases} 5.91 + 5.8 \times 10^{-3}\,T & 298 < T < 395 \\ 5.21 + 7.05 \times 10^{-3}\,T & 395 < T < 478 \\ 2.98 + 11.1 \times 10^{-3}\,T & 479 < T < 591 \\ 9.0 & 591 < T < 724 \\ 8.4 & 724 < T < 749 \\ 8 & 749 < T < T_{boil} \end{cases} \tag{4}$$

We will also need a reference point on which to base our plots. The molar enthalpy of formation for Pu is $\overline{H}(T = 298,\ P = 1\,atm) = 0\frac{cal}{mol}$ and the molar entropy of formation is $\overline{S}(T = 298,\ P = 1\,atm) = 13.2\frac{cal}{K\,mol}$. We also need to know the molar enthalpy for transformation for each phase change. These are provided in table 1.

Using this information, we will be able to determine the temperature functionality for the three thermodynamic potentials.

1 Enthalpy vs. Temperature for Pu

To determine the molar enthalpy of Pu as a function of temperature, we start by combining equ. (4) with (1). This is the same approach we have used for H_2O and carbon:

Tab. 1: Molar enthalpies of transformation for the stable solid phases of Pu.

phase	$\triangle \overline{H}$ $\left[\frac{K\,cal}{mol}\right]$	T_{eq}^{trans} $[K]$
$\alpha \rightarrow \beta$	800	395
$\beta \rightarrow \gamma$	150	478
$\gamma \rightarrow \delta$	130	591
$\delta \rightarrow \epsilon$	20	724
$\epsilon \rightarrow liquid$	440	749

$$\overline{H}(T) = \begin{cases} \overline{H}(298) + \int_{298}^{T} \overline{C}_p^{\alpha} dT & 298 < T < 395 \\ \overline{H}^{\alpha}(395) + \Delta \overline{H}_{\alpha \rightarrow \beta} + \int_{395}^{T} \overline{C}_p^{\beta} dT & 395 < T < 478 \\ \overline{H}^{\beta}(478) + \Delta \overline{H}_{\beta \rightarrow \gamma} + \int_{478}^{T} \overline{C}_p^{\gamma} dT & 479 < T < 591 \\ \overline{H}^{\gamma}(591) + \Delta \overline{H}_{\gamma \rightarrow \delta} + \int_{591}^{T} \overline{C}_p^{\delta} dT & 591 < T < 724 \\ \overline{H}^{\delta}(724) + \Delta \overline{H}_{\delta \rightarrow \epsilon} + \int_{724}^{T} \overline{C}_p^{\epsilon} dT & 724 < T < 749 \\ \overline{H}^{\epsilon}(749) + \Delta \overline{H}_{\epsilon \rightarrow liquid} + \int_{749}^{T} \overline{C}_p^{liquid} dT & 749 < T < T_{boil} \end{cases} \quad (5)$$

Keep in mind we have defined a reference point above.

The molar enthalpy of the stable phase is determined by integrating the heat capacity of the stable phase up to the transformation temperature. The enthalpy then "jumps" by the heat of transformation. We have defined our enthalpy curves in a very similar way to that of H_2O, only with more phases.

In Maxima, we first define our specific heat functions and enthalpies of transformation:

```
(%i1)   Cp_a(T):=5.91+5.8e-3*T$
(%i2)   Cp_b(T):=5.21+7.05e-3*T$
(%i3)   Cp_g(T):=2.98+11.1e-3*T$
(%i4)   Cp_d:9$ Cp_e:8.4$ Cp_l:8$
(%i7)   H_ab:800$ H_bg:150$ H_gd:130$ H_de:20$ H_el:440$
```

Notice the distiction between ":=" and ":". The former defines a function, while the later defines a value.

Next, we separately determine the enthalpy curves for each phase.

```
(%i12)  H1(T):=integrate(Cp_a(T),T,298,T)$ H1(T);
```

(%o13) $\frac{29\,T^2+59100\,T}{10000} - \frac{538996}{267}$

```
(%i14)  T:395$ H_a:%o13$ remvalue(T)$
(%i17)  H2(T):=H_a+H_ab+integrate(Cp_b(T),T,395,T)$ H2(T);
```

(%o18) $\frac{141\,T^2+208400\,T}{40000} - \frac{10771158469}{10359600}$

```
(%i19)  T:478$ H_b:%o18$ remvalue(T)$
```

```
(%i22)  H3(T):=H_b+H_bg+integrate(Cp_g(T),T,478,T)$ H3(T);
```

(%o23) $\frac{111\,T^2+59600\,T}{20000}-\frac{16285183326077}{56848305000}$

```
(%i24)  T:591$ H_g:%o23$ remvalue(T)$
(%i27)  H4(T):=H_g+H_gd+integrate(Cp_d,T,591,T)$ H4(T);
```

(%o28) $9\,(T-591)+\frac{805704705070543}{227393220000}$

```
(%i29)  T:724$ H_d:%o28$ remvalue(T)$
(%i32)  H5(T):=H_d+H_de+integrate(Cp_e,T,724,T)$ H5(T);
```

(%o33) $8.4\,(T-724)+\frac{1082442253810543}{227393220000}$

```
(%i34)  T:749$ H_e:%o33$ remvalue(T)$
(%i37)  H6(T):=H_e+H_el+integrate(Cp_l,T,749,T)$ H6(T);
```

(%o38) $8\,(T-749)+5410.222199283439$

To emulate equ. (5) within Maxima, we use an if-then statement:

```
(%i39)  H(T):=if T<395 then %o13 elseif T<478 then %o18 elseif T<591
        then %o23 elseif T<724 then %o28 elseif T<749 then %o33 else
        %o38$
```

Now we have a function for the enthalpy that we can plot versus temperature.

The transition $\delta \rightarrow \epsilon$ is hardly noticable on the plot, since $\triangle\overline{H}_{\delta\rightarrow\epsilon}$ is a relatively small number.

```
(%i40)  plot2d([H(T)], [T,298,1000], [xlabel,"Temperature (K)"], [yla-
        bel,"Molar Enthalpy (cal/mol)"]);
```

Figure 1 shows the resulting GNUplot output.

2 Gibbs Free Energy Curves for Solid Phases of Plutonium

We determine the entropy curve in a similar manner to the way we determined the enthalpy curve.

In order to determine Gibbs free energy curves from equ. 3, we first need to determine the molar entropies as functions of temperature. Using equ. 2 in combination with equ. 4 and our reference values listed above, we determine the entropy functions:

Fig. 1: Molar Enthalpy as a function of temperature for plutonium.



$$\overline{S}(T) = \begin{cases} \overline{S}(298) + \int_{298}^{T} \frac{\overline{C}_p^{\alpha}}{T} dT & 298 < T < 395 \\ \overline{S}^{\alpha}(395) + \Delta\overline{S}_{\alpha\to\beta} + \int_{395}^{T} \frac{\overline{C}_p^{\beta}}{T} dT & 395 < T < 478 \\ \overline{S}^{\beta}(478) + \Delta\overline{S}_{\beta\to\gamma} + \int_{478}^{T} \frac{\overline{C}_p^{\gamma}}{T} dT & 479 < T < 591 \\ \overline{S}^{\gamma}(591) + \Delta\overline{S}_{\gamma\to\delta} + \int_{591}^{T} \frac{\overline{C}_p^{\delta}}{T} dT & 591 < T < 724 \\ \overline{S}^{\delta}(724) + \Delta\overline{S}_{\delta\to\epsilon} + \int_{724}^{T} \frac{\overline{C}_p^{\epsilon}}{T} dT & 724 < T < 749 \\ \overline{S}^{\epsilon}(749) + \Delta\overline{S}_{\epsilon\to liquid} + \int_{749}^{T} \frac{\overline{C}_p^{liquid}}{T} dT & 749 < T < T_{boil} \end{cases} \tag{6}$$

The molar entropy of the stable phase is determined by integrating the heat capacity of the stable phase divided by the temperature up to the transformation temperature. Similar to enthalpy, the entropy then jumps by the entropy change during the transformation.

Once enthalpy and entropy are determined, then the Gibbs free energy curves are easy!

The entropy and enthalpy functions combined with equ. 3 provide us with the molar Gibbs free energy curves for the two solid phases of carbon:

$$\overline{G}(T) = \begin{cases} \overline{H}^{\alpha} - T\,\overline{S}^{\alpha} & 298 < T < 395 \\ \overline{H}^{\beta} - T\,\overline{S}^{\beta} & 395 < T < 478 \\ \overline{H}^{\gamma} - T\,\overline{S}^{\gamma} & 479 < T < 591 \\ \overline{H}^{\delta} - T\,\overline{S}^{\delta} & 591 < T < 724 \\ \overline{H}^{\epsilon} - T\,\overline{S}^{\epsilon} & 724 < T < 749 \\ \overline{H}^{liquid} - T\,\overline{S}^{liquid} & 749 < T < T_{boil} \end{cases} \tag{7}$$

In Maxima we first determine our entropy as a function of temperature curves, then we determine our Gibbs free energy curves.

Similarly to enthalpy, we determine the entropy curves for each phase.

```
(%i41)  S1(T):=13.2+integrate(Cp_a(T)/T,T,298,T); S1(T);
```

(%o42) $$\frac{29550\,log(T)+29\,T}{5000}-\frac{14775\,log(298)+4321}{2500}+13.2$$

```
(%i43)  T:395$ S_a:%o42$ remvalue(T)$
(%i46)  S2(T):=S_a+H_ab/395+integrate(Cp_b(T)/T,T,395,T)$ S2(T);
```

(%o47) $$\frac{104200\,log(T)+141\,T}{20000}+\frac{29550\,log(395)+11455}{5000}-\frac{20840\,log(395)+11139}{4000}-\frac{14775\,log(298)+4321}{2500}+15.2253164556962$$

```
(%i48)  T:478$ S_b:%o47$ remvalue(T)$
(%i51)  S3(T):=S_b+H_bg/478+integrate(Cp_g(T)/T,T,478,T)$ S3(T);
```

(%o52) $$\frac{29800\,log(T)+111\,T}{10000}+\frac{104200\,log(478)+67398}{20000}-\frac{14900\,log(478)+26529}{5000}+\frac{29550\,log(395)+11455}{5000}+\frac{-20840\,log(395)-11139}{4000}+\frac{-14775\,log(298)-4321}{2500}+15.53912398707695$$

```
(%i53)  T:591$ S_g:%o52$ remvalue(T)$
(%i56)  S4(T):=S_g+H_gd/591+integrate(Cp_d/T,T,591,T)$ S4(T);
```

(%o57) $$9\,(log\,(T)-log\,(591))+\frac{29800\,log(591)+65601}{10000}+\frac{104200\,log(478)+67398}{20000}+\frac{-14900\,log(478)-26529}{5000}+\frac{29550\,log(395)+11455}{5000}+\frac{-20840\,log(395)-11139}{4000}+\frac{-14775\,log(298)-4321}{2500}+15.7590901461294$$

```
(%i58)  T:724$ S_d:%o57$ remvalue(T)$
(%i61)  S5(T):=S_d+H_de/724+integrate(Cp_e/T,T,724,T)$ S5(T);
```

(%o62) $$8.4\,(log\,(T)-log\,(724))+9\,(log\,(724)-log\,(591))+\frac{29800\,log(591)+65601}{10000}+\frac{104200\,log(478)+67398}{20000}+\frac{-14900\,log(478)-26529}{5000}+\frac{29550\,log(395)+11455}{5000}+\frac{-20840\,log(395)-11139}{4000}+\frac{-14775\,log(298)-4321}{2500}+15.78671445552167$$

```
(%i63)  T:749$ S_e:%o62$ remvalue(T)$
(%i66)  S6(T):=S_e+H_el/749+integrate(Cp_l/T,T,749,T)$ S6(T);
```

(%o67) $$8\,(log\,(T)-log\,(749))+8.4\,(log\,(749)-log\,(724))+9\,(log\,(724)-log\,(591))+\frac{29800\,log(591)+65601}{10000}+\frac{104200\,log(478)+67398}{20000}+\frac{-14900\,log(478)-26529}{5000}+\frac{29550\,log(395)+11455}{5000}+\frac{-20840\,log(395)-11139}{4000}+\frac{-14775\,log(298)-4321}{2500}+16.374164388766$$

Fortunately, we have already calculated the molar enthalpies and entropies for the Pu system. Therefore, calculating the Gibbs free energy and plotting the curves is relatively straightforward.

```
(%i68)  G1(T):=H1(T)-T*S1(T);
```

Fig. 2: Molar Gibbs free energy as a function of temperature for Pu.




```
(%i69)    G2(T):=H2(T)-T*S2(T);
(%i70)    G3(T):=H3(T)-T*S3(T);
(%i71)    G4(T):=H4(T)-T*S4(T);
(%i72)    G5(T):=H5(T)-T*S5(T);
(%i73)    G6(T):=H6(T)-T*S6(T);
```

The plot option "legend" allows us to assign names to the various curves on the GNUplot output.

For the Gibbs free energy, we plot all of the curves so that we may see the intersection points. These points of intersection provide the transition temperatures, and the lowest Gibbs curve indicates the stable phase for the given temperature.

```
(%i74)    plot2d([G1(T), G2(T), G3(T), G4(T), G5(T), G6(T)], [T,298,1000],
          [xlabel,"Temperature (K)"],[ylabel,"Gibbs Free Energy (J/mol)"],
          [legend, "alpha", "beta", "gamma", "delta", "epsilon", "liquid"]);
```

Figure 2 shows the resulting GNUplot output. As expected, the intersections correspond to the transition temperatures for each phase change. Also, the lowest Gibbs curve for a particular temperature range corresponds to correct phase for that range.